

# SAS-Karten und die Verwendung in digitalen Medien und Druckerzeugnissen – Empfehlungen für Ausgabeformate und Einstellungsoptionen

Susanne Steinmann  
aQua - Institut für angewandte  
Qualitätsförderung und Forschung  
im Gesundheitswesen GmbH  
Maschmühlenweg 8-10  
37073 Göttingen  
Susanne.Steinmann@aQua-institut.de

Tanja Seeliger  
aQua - Institut für angewandte  
Qualitätsförderung und Forschung  
im Gesundheitswesen GmbH  
Maschmühlenweg 8-10  
37073 Göttingen  
Tanja.Seeliger@aQua-institut.de

Thomas G. Grobe  
aQua - Institut für angewandte  
Qualitätsförderung und Forschung  
im Gesundheitswesen GmbH  
Maschmühlenweg 8-10  
37073 Göttingen  
Thomas.Grobe@aQua-institut.de

## Zusammenfassung

Mit der SAS/GRAPH Prozedur GMAP können Daten mit geografischem Bezug in Form von annotierten Landkarten dargestellt werden. In der Regel werden die ausgegebenen Kartendarstellungen als Bildmaterial in digitale Medien oder Printmedien eingebunden oder mit gängigen Grafikbearbeitungsprogrammen weiterverarbeitet. Der Beitrag erläutert, welche Dateiformate und SAS-Ausgabe-Parameter für eine reibungslose Integration, Weiterverarbeitung und eine optimale Wiedergabe der Karten im Endprodukt empfohlen werden können.

**Schlüsselwörter:** SAS/GRAPH, PROC GMAP, SAS GfK GeoMarketing Map Data Germany, Dateiformate, OPTIONS, GOPTIONS, SAS ODS

## 1 Motivation

Die SAS-Prozedur GMAP des SAS/GRAPH-Moduls ermöglicht die Darstellung von Daten mit geografischem Bezug in Form von annotierten Landkarten. Diese sind in der Regel nicht für die Verwendung innerhalb des SAS-Systems selbst bestimmt, sondern werden zumeist in Medien wie Print-Produkten, Apps und Webseiten eingebunden. Für die Erstellung und Ausgabe der Landkarten bietet SAS eine Vielzahl verschiedener Lösungen. Die Auswahl geeigneter Einstellungen ist essentiell für die Qualität der Darstellung im Endprodukt und hat erheblichen Einfluss auf die Möglichkeiten und den Aufwand der Weiterverarbeitung. Dieser Beitrag will mit der Vorstellung geeigneter Da-

teiformate und Ausgabeeinstellungen für SAS-Kartendarstellungen Anregungen für die Optimierung den SAS-Karten-Output geben.

## 2 Vorgehensweise

Es wurden zunächst drei häufige Szenarien für die Verwendung der mit SAS erstellten annotierten Landkarten identifiziert:

- Grafische Weiterbearbeitung der Kartendarstellung in einer gängigen Software zur Grafikbearbeitung,
- Einbindung der Kartendarstellung (ohne weitere Bearbeitung) in ein Berichtsdokument sowie
- Nutzung der Kartendarstellung innerhalb einer Webpräsenz.


Als Grundlage für die Bewertung der annotierten Landkarten hinsichtlich Ihrer Eignung für die genannten Verwendungszwecke wurde mit SAS 9.4 eine Abbildung erstellt, die eine Reihe wesentlicher Elemente von SAS-Kartendarstellungen enthält (siehe Abbildung 1). Dabei diente der SAS GfK-Datensatz für Deutschland als Basis für die Erstellung einer Übersichtskarte für Deutschland mit eingezeichneten Kreisgrenzen. In der Karte wurden darüber hinaus Bundeslandgrenzen eingetragen. Hierfür wurden in Abhängigkeit von den Eigenschaften des Dateiformats der Zieldatei entweder ein ANNOTATE-Datensatz [1] oder übereinandergelegte Karten mit Transparenzfarben [2] genutzt. In der Karte wurden Städte, in denen die KSFE bisher stattgefunden hat, nach Angaben zu Längengrad und Breitengrad für die Postleitzahlen der Veranstaltungsorte<sup>1</sup> eingezeichnet. Die Größe der eingezeichneten Städte-Markierungen erfolgte in Abhängigkeit von der Anzahl der KSFE-Veranstaltungen, die der jeweiligen Stadt zuzuordnen waren. Die Städtenamen und die Veranstaltungsjahre wurden über den ANNOTATE-Datensatz in der Kartendarstellung vermerkt. Der vollständige Programmcode zur Erstellung einer entsprechenden annotierten Landkarte ist im Anhang dieses Beitrages aufgeführt.

Für die genannten Szenarien – grafische Weiterbearbeitung, die Einbindung in ein Berichtsdokument und die Nutzung innerhalb einer Webpräsenz – wurden typische Dateiformate gewählt und die erstellte Landkarte unter Verwendung unterschiedlicher Einstellungen und Optionen in diesen Dateiformaten ausgegeben. Die resultierenden Kartenabbildungen wurden im Hinblick auf ihre Eignung für den Verwendungszweck bewertet und falls nötig und möglich durch Anpassungen des SAS-Programmcodes optimiert. Auf dieser Grundlage werden Empfehlungen für geeignete Dateiformate und den entsprechenden SAS-Programmcode gegeben.

---

<sup>1</sup> Liste der KSFE-Veranstaltungsorte abrufbar unter:  
[http://de.saswiki.org/wiki/Konferenz\\_der\\_SAS-Anwender\\_in\\_Forschung\\_und\\_Entwicklung](http://de.saswiki.org/wiki/Konferenz_der_SAS-Anwender_in_Forschung_und_Entwicklung),  
Google Maps Koordinaten abrufbar unter: <https://www.gpskoordinaten.de/>

Die Erstellung der annotierten Landkarte erfolgte mit SAS 9.4 (M5). Tests zur Dateieignung wurden mit Adobe Illustrator CS6 (16.0.3), Inkscape 0.92.4, Microsoft Word 2013, Adobe Acrobat Professional 11.0.23 sowie Mozilla Firefox Nightly 60.4.0esr und Internet Explorer 11.0 unter Microsoft Windows 7 Enterprise 6.1 durchgeführt.



**Abbildung 1:** Deutschlandkarte mit KSFE-Veranstaltungsorten

Hinweis: Der SAS-GfK-Datensatz für Deutschland datiert vom Jahr 2012. Nicht berücksichtigt ist daher die Fusion der Landkreise Göttingen und Osterode am Harz von 1. November 2016. Die Stadtgemeinde Bremen ist in diesem Datensatz fälschlich dem Bundesland Niedersachsen zugeordnet, so dass das Bundesland Bremen bei einer Darstellung der Bundeslandgrenzen ausschließlich mit dem Gebiet Bremerhaven in der Karte eingezeichnet wird.

### 3 Wichtige Elemente des SAS-Karten-Outputs

Die Grafikausgabe aus den SAS Modulen SAS/BASE und SAS/GRAPH erfolgt auf unterschiedliche Weise. Während Grafiken aus SAS BASE über das OUTPUT DELIVERY SYSTEM (ODS) ausgegeben werden, wird die Ausgabe aus SAS/GRAPH mit Hilfe verschiedener Grafik-Treiber (Devices) prozessiert. Das ODS ist der SAS-Standard für die Erstellung von SAS-OUTPUT, mit dem verschiedene Ausgabeziele

(Destinations), angesteuert werden können. Die Standardeinstellung des ODS ist die Ausgabe nach HTML. Da die SAS-Prozedur PROC GMAP Bestandteil des SAS Moduls SAS/GRAPH ist, ist die Ausgabe der erzeugten Landkarte in eine eigenständige Datei durch die Angabe eines Grafik-Treibers möglich. Um Ausgabeort, Dateiformat, Erscheinungsbild und Eigenschaften der Kartendarstellung dem jeweiligen Verwendungszweck anzupassen, müssen in der Regel Kombinationen aus SAS BASE, SAS/GRAPH und ODS-Befehlen verwendet werden [3]. Eine entsprechende Syntax ist in Listing 1 beispielhaft aufgeführt und dient zur Erläuterung typischer SAS-Befehle. Die Nummerierung in der SAS-Syntax verweist auf die gleichermaßen nummerierten Erläuterungen. Empfehlungen von entsprechenden Einstellungen für die genannten Verwendungszwecke sind in Kapitel 4 aufgeführt.

**Listing 1:** Elemente des SAS-Programmcodes für die Karten-Ausgabe

```
(1)      * SAS System-Optionen;
        options nofontembedding;

        * Wahl der ODS Destination;
(2)      ods all close;
(3)      ods listing;

        * Pfad und Dateiname für Grafikausgabe;
(4)      filename gout "C:\temp\grafik1.jpg";

        * SAS Grafik-Optionen;
(5)      goptions
(6)          reset=all
(7)          hsize=12cm
(8)          vsize=15cm
(9)          xpixels=300
(10)         xmax=lin
(11)         ypixels=300
(12)         ymax=lin
(13)         notransparency
(14)         cback="CXFFFFFF"
(15)         ftext="arial"
(16)         gsfname=gout
(17)         gsfmode=replace
(18)         device=JPEG;

        * Angabe des Füllmusters;
(19)     pattern value=e;

        * Angabe des Titels;
(20)     title "Deutschland mit Kreisgrenzen";

        * PROC GMAP (Kartenerstellung);
(21)     proc gmap map=mapsgfk.germany data=mapsgfk.germany (obs=1) all;
        id id;
        choro id / nolegend;
        run;

        * Löschen des Titels;
(22)     title;

        * ODS Destination schließen und zurücksetzen;
(23)     ods listing close;
        ods html;
```

(1) (15) Die System-Option FONTEMBEDDING gibt an, dass die Einbettung von Schriften in die Ausgabedatei erfolgen soll. FONTEMBEDDING ist die Voreinstellung dieser Option, die mit Angabe von NOFONTEMBEDDING deaktiviert werden kann. Es können bei der Erstellung von Grafiken mit den SAS/GRAPH-Prozeduren verschiedene Fonts (deutsch: Schriften/Schriftfamilien) ausgewählt werden, die mit der SAS-Installation bereitgestellt wurden [3]. Weitere (proprietäre) Schriften können mit der Prozedur FONTREG in SAS registriert werden. Wird die mit SAS erstellte Grafikdatei weitergegeben, wird beim Öffnen der Datei auf die Schriftbibliothek des hierfür verwendeten Rechners zurückgegriffen. Die in der Abbildung enthaltene Schrift wird nur dann korrekt dargestellt, wenn sie dort ebenfalls unter gleichem Namen installiert oder in die Datei eingebettet ist. Anderenfalls wird die Schrift einstellungsabhängig ersetzt. Bei der Einbettung von Schriften ist zu berücksichtigen, dass die Nutzung der Option nicht von jeder Software unterstützt wird und entsprechende auch zu unerwarteten Ergebnissen bei der Weiterverwendung der Grafik-Datei führen kann.

(2) (3) (23) Die Kartenabbildung kann als eigenständige Datei oder eingebettet in Dateien wie Textdokumente oder Webseiten ausgegeben werden. Soll die Kartendarstellung nicht (nur) als HTML-Datei im Output-Viewer erscheinen, wird die entsprechende DESTINATION im ODS-Statement angegeben. Für eigenständige Grafikdateien ist dies zumeist die Destination LISTING. Die Ausgabe erscheint in allen geöffneten DESTINATIONS, weshalb DESTINATIONS, die nicht (mehr) benötigt werden, wieder geschlossen werden sollten.

(4) (16) (17) Standardmäßig wird die Ausgabe aus SAS/GRAPH Prozeduren im voreingestellten Katalog WORK.GSEG gespeichert. Der Dateiname folgt den GRSEG-Namenskonventionen. Mit dem FILENAME-Statement (4) kann ein abweichender Speicherort und Dateiname festgelegt werden. Die im FILENAME-Statement angegebene Referenz muss vor der Angabe der GOPTIONS erfolgen, damit sie dort mit der GSFNAME-Option aufgerufen werden kann. Die angegebene Dateiendung muss für die genutzte DEVICE gültig sein. Die Grafikoption GSFMODE (17) gibt an, ob bei erneuter Ausgabe von Grafiken unter demselben Grafiknamen (16) eine Ersetzung der bestehenden Datei (REPLACE), ein Hinzufügen zu der bestehenden Datei (APPEND) oder das Senden der Ausgabe zu einem mit der GACCESS Option genannten POST erfolgt.

(5) Das GOPTIONS-Statement kann als globales Statement an beliebiger Stelle im SAS-Programmcode ausgeführt werden. Es muss allerdings nach der Angabe eines spezifischen Dateinamens (4) und vor der Ausführung der Grafik-Prozedur genannt werden, für die die Einstellungen gelten sollen. Es werden im Weiteren nur die Optionen erläutert, die im Zusammenhang mit der Grafikausgabe der aufgeführten Beispieldateien genutzt werden. Eine vollständige Aufstellung mit Erläuterung der einzelnen GOPTIONS kann der SAS/GRAPH Referenz entnommen werden [3].

(6) Da die Einstellungen, die einmal über das GOPTIONS-Statement getroffen wurden während der SAS-Session aktiv bleiben, empfiehlt es sich zunächst die RESET-Option

aufzurufen, um möglicherweise noch aktive, aber nicht mehr benötigte Einstellungen zu deaktivieren. Während alle anderen GOPTIONS in beliebiger Reihenfolge genannt werden können, muss die Option RESET als erste Option aufgeführt werden. Die Werte einzelner Optionen können auch mit einem „Leer-Statement“ (z.B. CBACK= ) zurück gesetzt werden.

(7) (8) Für viele Dateiformate kann die Größe der auszugebenden Grafik mit der Angabe der horizontalen (HSIZE) und vertikalen (VSIZE) Abmessungen festgelegt werden. Die Angaben zu Größe und Auflösung der Grafik können jeweils in Zellen (cells), Zentimeter (cm), Inch (in), Prozent des Grafik-Outputs (pct) und Punkten (pt) vorgenommen werden. Abmessungen für Output im SVG-Format werden dagegen mit den System-Optionen SVGWIDTH und SVGHEIGHT im OPTIONS-Statement angegeben werden.

(9) (10) (11) (12) Die Bildauflösung – die Anzahl der Bildpunkte pro Spalte und Zeile – der Grafik wird in der Regel in der Einheit „dpi“ (dots per inch, deutsch: Punkte pro Inch) angegeben. Sie kann in Abhängigkeit von Dateiformat und Destination auf verschiedene Arten vorgegeben werden, von denen einige hier aufgeführt sind:

- Es kann eine DEVICE genutzt werden, die eine spezifische Bildauflösung beinhaltet (beispielweise die DEVICE „JPEG300“ zur Ausgabe einer JPEG-Datei mit 300 dpi).
- Bei der Ausgabe einer eigenständigen Grafik mit der Device im GOPTIONS Statement kann die Bildauflösung für viele Ausgabe-Formate mit der Angabe der Pixel auf der X-Achse der Grafik (XPIXEL) und der äußeren Abmessung der Grafik auf der X-Achse (XMAX) sowie der entsprechenden Angaben für die Y-Achse (YPIXEL, YMAX) vorgegeben werden. Die Auflösung ergibt sich aus der Anzahl der angegebenen Pixel pro Achse (also beispielsweise Auflösung der X-Achse = XPIXEL/XMAX). Die Angaben zur Dateiauflösung mit diesen Optionen überschreiben die Festlegung der Bildauflösung mit einer spezifischen DEVICE wie beispielsweise JPEG300.
- Bei der Ausgabe mit den Destinations ODS LISTING oder ODS HTML kann bei der Verwendung der Universal Printer PNG, UPNG, PNGT, PNG300, SVG, SVGT, SVGVIEW, SVGZ und UEMF im ODS-Statement die Angabe der Bildauflösung in DPI mit der Option IMAGE\_DPI= erfolgen.
- Verschiedene ODS Destinations erlauben die Angabe der Auflösung mit Hilfe der Option DPI.

Wird keine Auflösung explizit angegeben, wird auf die SAS-Voreinstellung für das Dateiformat zurückgegriffen.

(13) Wenn eine DEVICE zur Ausgabe genutzt wird, die Transparenzfarben darstellen kann (beispielsweise PNG, SVGT, GIF), kann mit der Angabe der Option

TRANSPARENCY ein transparenter Bildhintergrund gewählt werden. Die Voreinstellung für diese Option ist NOTRANSPARENCY.

(14) Farben für einige Grafik-Elemente können über die Farboptionen des GOPTIONS-Statements vergeben werden. Verwendet wird im vorliegenden Beispiel die Option CBACK (für die Farbe des Grafikhintergrunds). Es kann mit der Option COLORS auch eine Farbliste angegeben werden. Dies ist sinnvoll, wenn in der Grafik beispielsweise (abgestufte) Farben für bestimmte Wertebereiche in einer festgelegten Reihenfolge vergeben werden sollen.

Es stehen verschiedene Farbschemata zur Verfügung, die zur Vergabe von Farben für Abbildungselemente genutzt werden können. Die Angabe von Farben kann beispielweise mit Hilfe des SAS-internen Farbbenennungssystems oder unter Verwendung des RGB-, RGBA- oder CMYK-Farbraums erfolgen. Die Farbwerte werden als Hexadezimalzahlen angegeben. Aufgeführt sind hier einige Farbschemata, die bei den folgenden Beispielen zur Anwendung kommen.

**Tabelle 1:** Beispiele für die Angabe von Farbwerten in SAS

Farbschema	Farbangabe	Erläuterung
SAS Registry	blue	Farbliste und Bezeichnungen s. SAS-Graph Reference [3]
RGB Farbwerte	CX0000FF	CX, Wert Red, Wert Green, Wert Blue
RGBA Farbwerte	A0000FFF	A, Wert Red, Wert Green, Wert Blue, Wert Alpha*
CMYK Farbwerte	FFFF0000	Wert Cyan, Wert Magenta, Wert Yellow, Wert Black

\* Alpha Kanal für Transparenzwert (in Prozent der vollständigen Deckkraft von „00“ bis „FF“)

Innerhalb desselben Statements können Farbbenennungen unterschiedlicher Farbschemata verwendet werden. Bei der Angabe der Farbwerte ist zu berücksichtigen, welches Farbschema von dem gewählten Ausgabeformat unterstützt wird. Zur Ermittlung und Umrechnung der Farbwerte steht in SAS das Makro %COLORMAC zur Verfügung.

(15) Bei der Auswahl der verwendeten Schriften ist zwischen TrueType-Fonts und Bitmap-Fonts zu unterscheiden. TrueType-Fonts werden, ähnlich wie Vektor-Grafiken, als Konturen beschrieben und können daher verlustfrei beliebig größenskaliert werden, während zur Darstellung von Bitmap-Fonts jeder einzelne Bildpunkt beschrieben wird, wodurch eine nachträgliche Skalierung nicht möglich ist.

In Abhängigkeit von den bei der Installation gewählten Einstellungen bietet SAS verschiedene TrueType-Schriften, die von Windows einfach ersetzt werden können [3]:

- Albany AMT (Arial)
- Thorndale AMT (Times New Roman)
- Cumberland AMT (Courier)

(18) Der mit der Option `DEVICE` angegebene (Grafik-Treiber) bestimmt, in welchem Dateiformat die ausgegebene Grafik gespeichert wird. Bei der Grafik-Ausgabe mit dem `ODS` wird die Device im `ODS Statement` angegeben.

(19) Das `PATTERN Statement` enthält Angaben zu Füllfarbe und Füllmuster für die `SAS/GRAPH` Prozeduren.

(20) (22) Das `TITLE Statement` zur Vergabe eines (Abbildungs-)Titels kann an beliebiger Stelle im Programmcode aufgeführt werden (20). Der angegebenen Text wird solange verwendet, bis der Titel-Text ersetzt beziehungsweise zurückgesetzt wird (22).

(21) Nachdem alle Einstellungen getroffen sind, folgt die `SAS-Prozedur GMAP`, mit der die Kartendarstellung erstellt wird.

## **4 Empfehlungen für den SAS-Karten-Output**

### **4.1 Weiterverarbeitung des SAS-Karten-Outputs durch eine Grafikagentur**

An der Erstellung von umfangreichen Berichten und Publikationen sind häufig Agenturen beteiligt, die die grafische Umsetzung von Texten, Tabellen und Abbildungen übernehmen. Die beteiligte Agentur sollte daher die annotierte Landkarte zur Weiterverarbeitung in einem Dateiformat erhalten, das es ermöglicht, die Karte und einzelne Elemente daraus zu bearbeiten.

#### **4.1.1 Anforderungen**

Wichtige Kriterien zur Bewertung der Grafikausgabe für diesen Verwendungszweck sind:

- Die Grafikelemente sind sinnvoll gruppiert und einzeln auswählbar.
- Die Textelemente sind als Pfad oder Textfeld in der Datei enthalten.
- Die Schriftart wird wie im `SAS-Programm` angegeben in die Grafik übernommen.
- Exakte Farbwerte werden wie in `SAS` definiert in die Grafik übernommen oder können problemlos ersetzt werden.
- Die Abbildung ist größenskalierbar.
- Die Dateigröße ist handhabbar.

#### **4.1.2 Geeignete Dateiformate**

`SVG (Scalable Vector Graphics)`- und `EPS (Encapsulated PostScript)`-Dateien sind als Vektorgrafiken verlustfrei größenskalierbar. Sie können mit gängiger Grafikbearbeitungssoftware geöffnet und editiert werden. In entsprechenden Grafikdateien können einzelne Abbildungselemente getrennt als bearbeitbare Ebenen oder Pfade eingebunden werden. Die Dateiformate bieten eine korrekte Wiedergabe der angegebenen

Farbwerte. Die einzelnen Bildpunkte werden über Funktionen beschrieben und nicht als Pixel wiedergegeben, was sich positiv auf die Dateigröße auswirkt. Während Inhalte von SVG-Dateien über XML-Code beschrieben werden und sie sich auch zur Einbindung auf Webseiten eignen, finden EPS-Dateien besonders bei Printmedien Verwendung.

### 4.1.3 SAS-Programmcode, Hinweise und Bewertung

#### Listing 2: DEVICE-basierte Ausgabe einer SVG-Datei zur Weiterverarbeitung durch eine Grafikagentur

<code>options nofontembedding</code>	nicht jede Software unterstützt das Einbetten von Schriften, daher ist der Einsatz der Einstellung in Abhängigkeit von der Software zur Grafikbearbeitung zu erwägen (Alternative: FONTEMBEDDING)
<code>/*svgwidth=20cm svgheight=25cm*/;</code>	falls gewünscht Größe der Grafik angeben (für SVG-Dateien im OPTIONS-Statement), in der Regel bei diesem skalierbaren Grafik-Format nicht erforderlich
<code>ods _all_ close;</code> <code>ods listing</code> <code>/*image_dpi=300*/;</code>	ODS Destination LISTING wählen falls gewünscht Auflösung angeben, in der Regel bei diesem skalierbaren Grafik-Format nicht erforderlich
<code>filename gout</code> <code>"C:\temp\Grafik_SVG1.svg";</code>	Ausgabepfad und Dateinamen angeben
<code>goptions reset=all</code> <code>cback="CXFFFFFFF"</code> <code>ftext="arial"</code> <code>gsfname=gout gsfmode=replace</code>	Farben (RGB-Modus) angeben, Schrift, etc. angeben, Referenz aus dem FILENAME-Statement für die Dateiausgabe,
<code>device=svg;</code>	DEVICE=SVG angeben
<code>pattern value=ms color="CXEBEBEB"</code> <code>repeat=50;</code>	Füllmuster und Farbe angeben (Farbe im RGB-Modus)
<code>PROC GMAP ... QUIT;</code>	Prozedur GMAP
<code>ods listing close;</code> <code>ods html;</code>	ODS Destination schließen ODS Default wieder herstellen

### Listing 3: Ausgabe einer SVG-Datei zur Weiterverarbeitung durch eine Grafikagentur über ODS Printer

<pre>options nofontembedding printerpath=svg</pre>	mit der Option PRINTERPATH=SVG das Ausgabeformat angeben,
<pre>/*svgwidth=20cm svgheight=25cm*/</pre>	falls gewünscht Auflösung angeben, in der Regel bei diesem skalierbaren Grafik-Format nicht erforderlich,
<pre>nodate nonumber;</pre>	weitere Option (hier: kein Datum, keine Seitenzahl)
<pre>ods graphics / imagemap=off;</pre>	ToolTips, wenn nicht gewünscht deaktivieren (keine ungepflegten Mouse-Over-Effekte einbinden)
<pre>ods _all_ close; ods printer file="C:\temp\Grafik_SVG2.svg" /*dpi=300*/;</pre>	ODS Destination PRINTER wählen, Ausgabepfad und Dateinamen angeben, falls gewünscht Auflösung angeben
<pre>goptions reset=all cback="CXFFFFFFF" ftext="arial";</pre>	Farben (RGB-Modus) angeben, Schrift etc. angeben
<pre>pattern value=ms color="CXEBEBEB" repeat=50;</pre>	Füllmuster und Farbe angeben (Farbe im RGB-Modus)
<pre>PROC GMAP ... QUIT;</pre>	Prozedur GMAP
<pre>ods printer close; ods html; ods graphics / imagemap=on;</pre>	ODS Destination schließen ODS Default wieder herstellen ODS Graphics Default wieder herstellen

### Hinweise und Bewertung:

Auch wenn die Erstellung einer Datei im SVG-Format ebenso DEVICE-basiert über die DESTINATION LISTING möglich ist (s. Listing 2), hat die Erstellung mit Hilfe des ODS Printers einige Vorteile, die sich aus der Nutzung von Funktionen des ODS ergeben. So können einige Grafikelemente wie ToolTips besser kontrolliert werden. Im Prinzip erlauben beide Ausgabewege für Dateien im SVG-Format die weitere Bearbeitung in Standard-Software zur Grafikbearbeitung wie dem Adobe Illustrator, da die Grafikelemente getrennt ansprechbar sind und Gebietsgrenzen als geschlossene Polygone dargestellt werden. Allerdings werden Gebiete, die aus mehreren Polygonen bestehen, aber inhaltlich eine Einheit bilden (wie zum Beispiel ein Bundesland mit den dazugehörigen Inseln) bei der SAS-Ausgabe nicht unbedingt sinnvoll gruppiert, so dass hier ggf. eine geringe Nachbearbeitung erforderlich ist.

Ein besonderes Augenmerk ist auf die Annotation der ausgegebenen Grafik zu richten. Schnell führt die Einzeichnung von zusätzlichen Elementen wie Gebietsgrenzen mit Hilfe eines ANNOTATE-Datensatzes dazu, dass diese Einträge (wie z.B. die Grenzen

verschiedener Bundesländer) als ein einziges Grafikelement betrachtet werden. Es ist im entsprechenden Fall auf die saubere Verwendung geschlossener einzelner Polygone zu achten. Eine gute Alternative zur Einzeichnung verschiedener Gebietsgrenzen (wie Kreis- und Bundeslandgrenzen) bietet die Device SVGT (siehe SAS-Programmcode im Anhang), die es ermöglicht, mehrere Abbildungen mit unterschiedlichen Grenzverläufen als transparente Grafikelementen „übereinander zu legen“.

Farbwerte werden in SVG-Dateien unter Verwendung des RGB-Farbschemas verarbeitet. Die gewünschten Farben können bereits in SAS im RGB-Modus angegeben werden und werden korrekt in die Grafikdatei übernommen.

Dateien, die zahlreiche Abbildungselemente enthalten, benötigen unter Umständen viel Speicherplatz. Die DEVICE SVGZ bietet die Möglichkeit, komprimierte SVG-Dateien auszugeben. Entsprechende Dateien können allerdings in Web-Browsern nicht direkt geöffnet werden, bieten sich aber für eine Weiterverarbeitung an.

#### **Listing 4:** Ausgabe einer EPS-Datei zur Weiterverarbeitung durch eine Grafikagentur über ODS Printer

<pre>options nofontembedding printerpath=PSLEPSFC  /*papersize=(12cm 15cm)*;/  ods _all_ close; ods printer ps file="C:\temp\Grafik_EPS1.eps"  /*dpi=300*/;  goptions reset=all ftext="arial" ctitle="000000BF" ctext="000000BF" cback="00000000";  pattern value=ms color="00000014" repeat=50;  PROC GMAP ... QUIT;  ods printer close; ods html;</pre>	<pre>mit der Option PRINTERPATH= PSLEPSFC den Ausgabeweg angeben, (Device PSLEPSFC [Farbe]oder PSLEPSF [Grautöne]), falls gewünscht Größe angeben  ODS Destination PRINTER POSTSCRIPT wählen, Ausgabepfad und Dateinamen angeben, falls gewünscht Auflösung angeben, in der Regel bei diesem skalierbaren Gra- fik-Format nicht erforderlich  Schrift angeben, Farben (CMYK-Modus) etc. angeben  Füllmuster und Farbe angeben (Farbe im CMYK-Modus)  Prozedur GMAP, (Farbe auch hier im CMYK-Modus)  ODS Destination schließen ODS Default wieder herstellen</pre>
---	--

## Hinweise und Bewertung:

Auch in der erstellten EPS-Datei können einzelne Bildelemente ausgewählt und editiert werden. EPS-Dateien unterstützen das CMYK-Farbschema. Die Farbwerte können bereits in SAS unter Verwendung des CMYK-Farbschemas erfolgen und werden korrekt in die Datei-Ausgabe übernommen, was für die Umsetzung eines bestimmten Corporate Designs wichtig sein kann.

Die Verwendung der Option FONTEMBEDDING ist in hohem Maße von den Eigenschaften der verwendeten Grafikbearbeitungssoftware abhängig. Bei unseren Tests, die mit dem Adobe Illustrator CS6 (16.0.3) durchgeführt wurden, zeigte sich, dass bei Verwendung der OPTION NOFONTEMBEDDING Schrift in Textfelder und bei Angabe der OPTION FONTEMBEDDING Schrift in Pfade umgewandelt wurde. Welche Variante benötigt wird, ist im Anwendungsfall zu entscheiden.

## **4.2 Einbindung des SAS-Karten-Outputs in Berichtsdokumente**

In vielen Fällen dienen Kartendarstellungen zur Veranschaulichung von Sachverhalten, die in Berichtsdokumenten und Publikationen beschrieben werden. Der SAS-Karten-Output ist als eine Abbildung in dem entsprechenden Dokument vorgesehen und soll in dieses eingefügt werden. Im Wesentlichen gibt es die Möglichkeit, eine Grafik-Datei händisch in ein Berichtsdokument einzufügen oder automatisierte Berichte zu erzeugen, bei denen die Grafik direkt aus SAS in ein Berichtsdokument ausgegeben wird.

### **4.2.1 Anforderungen**

Wichtige Kriterien zur Bewertung der Grafikausgabe für diesen Verwendungszweck sind:

- Die Abbildung kann verlustfrei in das Textdokument eingefügt werden.
- Die Abbildungsmaße werden korrekt übernommen.
- Eine aus dem Textdokument erzeugte PDF-Datei gibt die Abbildung auch bei Vergrößerung der Ansicht zufriedenstellend wieder.
- Die Schrift wird wie im SAS-Programm angegeben in die Grafik übernommen.
- Exakte Farbwerte werden wie in SAS definiert in die Grafik übernommen
- Die Dateigröße ist handhabbar.

### **4.2.2 Geeignete Dateiformate**

Sollen zunächst eigenständige Grafik-Dateien für Microsoft Word Berichte erstellt werden, bietet sich das Format Windows Enhanced Metafile (EMF) von Microsoft an. EMF-Dateien sind größenskalierbar und können sowohl Vektor- und Raster-Grafikelemente enthalten. Das Datei-Format eignet sich besonders für die Einbindung von Vektorgrafik-Dateien in Microsoft Word Dokumente.

Alternativ besteht die Möglichkeit aus SAS heraus eine Datei im Rich Text-Format (RTF)-Format zu erstellen, die auch eingebettete Kartenabbildung im EMF-Format enthält. Das RTF-Format wurde von Microsoft für den Austausch von Dokumente zwischen verschiedenen Textverarbeitungsprogrammen entwickelt. Dateien im RTF-Format können sowohl Text als auch Angaben zu dessen Formatierung sowie weitere Elemente wie Tabellen und Abbildungen enthalten. Neben der Kartenabbildung könnten also über das ODS auch weitere Inhalte in ein RTF-Dokument geschrieben werden. Das Dokument kann danach in einer Textverarbeitungssoftware wie Microsoft Word geöffnet und weiterbearbeitet werden.

Nach der Fertigstellung eines Berichtsdokumentes wird dieses zumeist im Portable Document Format (PDF)-Format gespeichert und an die jeweiligen Adressaten weitergegeben. Das PDF-Format ist ein plattformunabhängiges Format, das den Austausch und die originalgetreue Darstellung von Dokumenten unabhängig von der Anwendung, mit der das Dokument ursprünglich erzeugt wurde, ermöglicht. Es stellt derzeit einen Standard für die Weitergabe von Berichtsdokumenten dar.

### 4.2.3 SAS-Programmcode, Hinweise und Bewertung

#### Listing 5: DEVICE-basierte Ausgabe einer EMF-Datei zur Einbindung in ein Berichtsdokument

<code>options nofontembedding;</code>	fontembedding wird nicht unterstützt
<code>ods _all_ close;</code>	
<code>ods listing</code>	ODS Destination LISTING wählen
<code>image_dpi=300;</code>	gewünschte Auflösung angeben (default 96 dpi)
<code>filename gout</code>	Ausgabepfad und Dateinamen angeben
<code>"C:\temp\Grafik_EMF1.emf";</code>	
<code>goptions reset=goptions</code>	
<code>ftext="arial"</code>	Schriftart für Text auswählen,
<code>hsize=12cm vsize=15cm</code>	Größe der Grafik angeben,
<code>cback="CXFFFFFFF"</code>	Farben für Grafikelemente festlegen,
<code>device=emf;</code>	Device EMF oder SASEMF auswählen
<code>pattern value=ms color="CXEBEBEB"</code>	Füllmuster und Farbe angeben (Farbe im
<code>repeat=50;</code>	RGB-Modus)
<code>PROC GMAP ... QUIT;</code>	Prozedur GMAP
<code>ods listing close;</code>	ODS Destination schließen
<code>ods html;</code>	ODS Default wieder herstellen

## Hinweise und Bewertung:

EMF-Dateien sind Vektorgrafiken und damit größenskalierbar. Sie werden von SAS mit der Standardauflösung 96 dpi ausgegeben, soweit keine andere Einstellung getroffen wurde. Die Auflösung ist über das ODS Statement einstellbar. Die Dateigröße nimmt bei Angabe einer höheren Auflösung zu, so dass keine unnötig hohe Auflösung gewählt werden sollte. Anders als z.B. Dateien im SVG-Format benötigen EMF-Dateien bereits bei der Ausgabe aus SAS relativ viel Speicherplatz. In EMF-Dateien können nur True-Type-Fonts eingebunden werden. FONTEMBEDDING wird nicht unterstützt.

### Listing 6: Ausgabe einer RTF-Datei als Berichtsdokument mit ODS RTF

<code>options papersize=A4</code>	Größe der Seite angeben,
<code>nofontembedding;</code>	weitere Optionen (fontembedding wird nicht unterstützt)
<code>ods _all_ close;</code>	
<code>ods rtf file="C:\temp\Grafik_RTF1.rtf"</code>	ODS Destination RTF aufrufen, Ausgabepfad und Dateinamen angeben, Auflösung angeben
<code>image_dpi=300;</code>	
<code>goptions reset=goptions</code>	
<code>ftext="arial"</code>	Schriftart für Text auswählen,
<code>hsize=12cm vsize=15cm</code>	Größe der Grafik angeben,
<code>cback="CXFFFFFF"</code>	Farben für Grafikelemente festlegen,
<code>device=sasemf;</code>	Device SASEMF auswählen (entspricht der Voreinstellung für ODS RTF)
<code>pattern value=ms color="CXEBEBEB"</code>	Füllmuster und Farbe angeben (Farbe im RGB-Modus)
<code>repeat=50;</code>	
<code>PROC GMAP ... QUIT;</code>	Prozedur GMAP
<code>ods rtf close;</code>	ODS Destination schließen
<code>ods html;</code>	ODS Default wieder herstellen

## Hinweise und Bewertung:

Die Ausgabe von EMF-Dateien mit SAS ist gut für die Verwendung in typischen Berichtsdokumenten geeignet. Obwohl es möglich ist, die mit SAS erstellten EMF-Grafiken in Microsoft Word zu bearbeiten, empfiehlt es sich, erst die fertig erstellte Grafik in Word einzufügen, da die Nachbearbeitung mit einer Reihe von Schwierigkeiten behaftet sein kann. Die Erstellung einer RTF-Datei mit SAS, in die eine Abbildung im EMF-Format eingebettet wird, schafft darüber hinaus die Voraussetzungen für eine automatisierte Berichtserstellung, zumal über die ODS Destination ODS RTF auch weitere Berichtselemente wie Texte und Tabellen ausgegeben werden können. Die Größe des Speicherplatzes der mit SAS erstellten RTF-Dateien kann schnell erhebliche Ausmaße annehmen. Nach dem Öffnen der Datei in Microsoft Word und dem Speichern im DOCX-Format, reduziert sich die Dateigröße allerdings wieder deutlich. Werden die

erstellten Microsoft Word Berichtsdokumente in PDF-Dateien umgewandelt, zeigen die eingefügten oder eingebetteten EMF-Dateien auch im PDF-Dokument eine ansprechende Auflösung und gute Skalierbarkeit.

### Listing 7: Ausgabe einer PDF-Datei als Berichtsdokument mit ODS PDF

<code>options papersize=A4</code>	Größe der Seite angeben
<code>fontembedding</code>	Schrift einbetten, besonders bei proprietären Schriften wichtig
<code>nodate nonumber;</code>	und ggf. weitere Option (hier: kein Datum, keine Seitenzahl)
<code>ods _all_ close;</code>	
<code>ods pdf</code>	ODS PDF aufrufen, Ausgabepfad und Dateinamen angeben,
<code>file="C:\temp\Grafik_PDF1.pdf"</code>	kein Inhaltsverzeichnis erstellen, falls nicht gewünscht
<code>notoc</code>	
<code>dpi=300;</code>	Auflösung angeben
<code>goptions reset=all</code>	
<code>ftext="arial"</code>	Schrift angeben
<code>hsize=12cm vsize=15cm</code>	Größe der Grafik angeben
<code>cback="CXFFFFFFF"</code>	Farbe des Hintergrunds und ggf. weitere Farben festlegen,
<code>device= SASEMF;</code>	SASPRTC ist die Voreinstellung für ODS PDF wenn keine Device angegeben wird (in unserem Beispiel nahezu kein optischer Unterschied)
<code>pattern value=ms color="CXEBEBEB"</code>	Füllmuster und Farben angeben (Farbe im RGB-Modus)
<code>repeat=50;</code>	
<code>PROC GMAP ... QUIT;</code>	Prozedur GMAP
<code>ods pdf close;</code>	ODS Destination schließen
<code>ods html;</code>	ODS Default wieder herstellen

### Hinweise und Bewertung:

Mit der Destination ODS PDF können die erstellten Grafiken und weitere Berichtselemente direkt in eine PDF-Datei geschrieben werden. Die eingebundene Abbildung erscheint ohne Qualitätsverlust im Ziel-Format und ist dort über die Zoom-Ansicht größenskalierbar. PDF-Dateien können mit SAS auch auf andere Weise ausgegeben werden, beispielweise über ODS Printer. Für den hier vorgestellten Zweck haben sich keine Vorteile eines dieser Ausgabewege gezeigt. In Abhängigkeit von anderen Elementen und Anforderungen an die zu erstellenden Berichte könnten sich aber auch weitere Ausgabemöglichkeiten als sinnvoll erweisen. Je mehr unterschiedliche formatierte Elemente Eingang in den zu erstellenden Bericht finden sollen, desto sinnvoller erscheint die Grafik-Ausgabe mit ODS PDF.

### **4.3 Einbindung des SAS-Karten-Outputs in einer Webseite**

Eine weitere Einsatzmöglichkeit von mit SAS erstellten Karten wäre die Einbindung in einer Webseite oder App. Auch könnten ganze Webseiten mit SAS ausgegeben werden, von denen die einzelne Kartendarstellung nur ein Bestandteil wäre. Bei den Überlegungen zu geeigneten Ausgabeformaten und Einstellungen stand bei den im Folgenden dargestellten Möglichkeiten die Erstellung einer Grafik-Datei im Vordergrund, die in eine (bestehenden) Webseite eingebunden werden könnte, um in einem begleitenden Text erläuterte Sachverhalte grafisch darzustellen.

#### **4.3.1 Anforderungen**

Wichtige Kriterien zur Bewertung der Grafikausgabe für diesen Verwendungszweck sind:

- Die Abbildung kann problemlos in Webseite integriert werden.
- Die Abbildungsmaße werden wie gewünscht umgesetzt.
- Die Abbildung ist problemlos skalierbar.
- Die Schrift wird wie im SAS-Programm angegeben in die Grafik übernommen.
- Exakte Farbwerte werden wie in SAS definiert in die Grafik übernommen.
- Der Aufbau der Grafik wird schnell und ohne Wartezeiten umgesetzt.
- Die Dateigröße ist handhabbar.

#### **4.3.2 Geeignete Dateiformate**

Mit den eingangs bereits vorgestellten SVG-Grafiken lassen sich die oben genannten Anforderungen gut erfüllen. Dabei ständen bei der Einbindung der Grafikdatei auf einer Webseite andere Eigenschaften als bei der Erstellung von SVG-Grafiken zur Bearbeitung durch eine Grafikagentur im Vordergrund. Wichtige Merkmale einer für das Web optimierten Datei sind unter 4.3.1 genannt. Diese sind durch die Wahl der entsprechenden Ausgabe-Einstellungen zu erreichen. Des Weiteren stehen in SAS zusätzliche SVG-Devices für transparente Abbildungselemente und Grafiken mit einfachen interaktiven Elementen zur Verfügung.

Als Alternative zum SVG-Format wird ergänzend die Erstellung von SAS-Output im Portable Network Graphic (PNG)-Format sowie im JPEG-Format (der Entwicklung der Joint Photographic Experts Group) vorgestellt. Bei beiden Dateiformaten handelt es sich um Raster-Formate, die ihre Stärken in der Darstellung von Farben und Farbverläufen haben und für Grafiken mit entsprechenden Elementen alternativ zum SVG-Format eingesetzt werden könnten.

### 4.3.3 SAS-Programmcode, Hinweise und Bewertung

#### Listing 8: Device-basierte Ausgabe einer PNG-Datei zur Einbindung in eine Webseite

options nofontembedding;	fontembedding wird von einigen Browsern nicht unterstützt
ods _all_ close;	
ods listing	ODS Destination LISTING,
image_dpi=300;	falls gewünscht Auflösung angeben
filename gout	Ausgabepfad und Dateinamen angeben
"C:\temp\Grafik_PNG1.png";	
goptions reset=all	
ftext="arial"	Schrift angeben,
/*hsize=12cm vsize=15cm*/	Größe der Abbildung angeben, wenn gewünscht
transparency	Transparenz angeben, wenn gewünscht, wird von PNG-Format unterstützt, die Grafik erscheint ohne Hintergrundfarbe, alternativ kann auch Device=PNGT gewählt werden,
ctitle="AFFFFFFF14"	Farben angeben (ggf. mit Transparenz-Angabe im RGBA-Modus möglich),
gsfname=gout gsfmode=replace	Referenz aus dem FILENAME-Statement für die Dateiausgabe, GSFMODE und Device angeben
device=png;	
pattern value=ms color="AEBEBEB00"	Füllmuster und Farben angeben (ggf. mit Transparenz-Angabe im RGBA-Modus möglich)
repeat=50;	
PROC GMAP ... QUIT;	Prozedur GMAP
ods listing close;	ODS Destination schließen
ods html;	ODS Default wieder herstellen

#### Hinweise und Bewertung:

Die Anforderungen für die Einbindung einer Grafik-Datei auf einer Web-Seite sind bereits durch die im Kapitel 4.1 beschriebene Ausgabe einer Abbildung im SVG-Format mit leicht angepassten Einstellungen gut erfüllbar, welche hier aus Platzgründen nicht wiederholt aufgeführt wurde. Der Vollständigkeit halber soll als mögliche Alternative für diesen Verwendungszweck auch die Erstellung einer Portable Network Graphic (PNG)-Datei aufgezeigt werden. Es ist ein Rastergrafikformat, das eine verlustfreie Kompression ermöglicht, unterschiedlicher Farbtiefen bietet und mit Transparenz arbeiten kann. Da der Hintergrund einer PNG-Datei komplett transparent ist, ist das Erscheinungsbild der Abbildung von der Hintergrund-Einstellung des Browsers, in dem die Datei geöffnet wird, abhängig. Das Datei-Format kann also eingesetzt werden, wenn es gewünscht wird, dass ein bestimmter Hintergrund durch die Abbildung hindurch sicht-

bar ist. Auch können Schriftfarben mit Transparenzwerten angegeben werden, wenn das dahinter liegende Bild sichtbar bleiben soll.

### Listing 9: Device-basierte Ausgabe einer JPEG-Datei zur Einbindung in eine Webpräsenz

```
options nofontembedding;

ods _all_ close;
ods listing;                                Öffnen der ODS Destination LISTING

filename gout                                Ausgabepfad und Dateinamen angeben
"C:\temp\Grafik_JPEG1.jpg";

goptions reset=all
hsize=12cm vsize=15cm                       Größe der Abbildung,
xpixels=1000 xmax=5in                       Auflösung,
ypixels=1200 ymax=6in
cback="CXFFFFFFF"                           Farben angeben (RGB-Modus),
ftext="arial"                                Schrift angeben,
gsfname=gout gsfmode=replace                Referenz aus dem FILENAME-Statement für
device=JPEG;                                die Dateiausgabe, GSFMODE und Device
                                           angeben (alternativ auch
                                           Device=JPEG300, mit einer Auflösung von
                                           300 dpi, ohne Angabe von xpixel, xmax,
                                           ypixel, ymax)

pattern value=ms color="CXEBEBEB"           Füllmuster und Farben angeben (Farbe im
repeat=50;                                  RGB-Modus)

PROC GMAP ... QUIT;                          Prozedur GMAP

ods listing close;                           ODS Destination schließen
ods html;                                    ODS Default wieder herstellen
```

### Hinweise und Bewertung:

Die besondere Stärke des JPEG-Dateiformates liegt in der Darstellung von Bildern mit Farbverläufe. Da entsprechende Eigenschaften bei der Ausgabe typischer Kartendarstellungen in der Regel nicht benötigt werden, kommt dem Format für die Karten-Ausgabe eine untergeordnete Bedeutung zu. Vorstellbar wäre aber, dass bei der Verwendung einer Fotografie als Hintergrundbild eine Ausgabe der Kartendarstellung im JPEG-Format sinnvoll sein könnte. Da es sich bei JPEG-Dateien um Pixel-Dateien handelt, ist es bei Nutzung des Dateiformates wichtig eine ausreichende Auflösung (für den Einsatz im Web-Bereich beispielsweise 200 dpi) anzugeben. Soll eine Abbildung im JPEG-Format im Printbereich verwendet werden, ist eine deutlich höhere Auflösung von mindestens 300 dpi angebracht.

## 5 Fazit

Es gibt eine Reihe unterschiedlicher Vorgehensweisen Daten mit geografischem Bezug mit SAS darzustellen und auszugeben. Den Autoren lag daran zu zeigen, dass Überlegungen zum Verwendungszweck und geeigneten Datei-Formaten der Ausgabe dieser Bilddateien vorausgehen sollten, da die zur Verfügung stehenden Optionen nur auf diese Weise sinnvoll eingrenzt werden können. Neben einer kurzen Beschreibung der Vorteile und Anwendungsbereiche ausgewählter Dateiformate wird im vorliegenden Beitrag auch der SAS-Programmcode zur Ausgabe entsprechender Bilddateien aufgeführt. Sowohl für die ausgewählten Dateiformate, vor allem aber auch für den vorgestellten Programmcode, existieren zweifellos sinnvolle Alternativen, sodass die vorliegende Aufstellung problemlos erweiterbar wäre. Für die Beurteilung der Eignung der erstellten Dateien für einen bestimmten Verwendungszweck sind neben SAS-internen Prozessen eine Reihe weiterer Faktoren einzubeziehen (wie beispielsweise die Software, mit der der SAS-Output im Nachgang verarbeitet wird, und das Betriebssystem unter dem die Verarbeitung durchgeführt wird). Bei der Erstellung dieses Beitrages konnte daher nur ein geringer Anteil der denkbaren Software-Konstellationen abgedeckt werden.

Unabhängig von diesen Limitationen können die aufgeführten Programmbeispiele zeigen, wie die Ausgabe von Kartendarstellungen mit SAS in den empfohlenen Dateiformaten auf relativ einfache Art und Weise realisiert werden kann und über die konkret aufgeführten Beispiele hinaus Denkanstöße zur Optimierung der Grafikausgabe geben.

## Literatur

- [1] Eberhart, M.: Make the Map You Want with PROC GMAP and the Annotate Facility. Proceedings NESUG 2008
- [2] Greiner, S., Tambascia, N.: Scalable Vector Graphics in SAS 9.2. Proceedings KSFE 2011
- [3] SAS Institute Inc. 2016. SAS/GRAPH® 9.4: Reference, Fifth Edition. Cary, NC: SAS Institute Inc.

## Anhang:

### SAS-Programmcode zur Erstellung des Kartenbeispiels als SVG-Grafik mit Transparenz und Mouse-Over-Funktion (Den Programmcode bitte in den Programmeditor von SAS einfügen.)

```
* KSFE 2019 - Steinmann, Seeliger, Grobe: SAS-Karten und die Verwendung in digitalen Medien und Druck-
erzeugnissen;
* Kontakt: susanne.steinmann@aquainstitut.de;

* Notwendige Anpassungen;
* Makrovariablen für Ausgabeverzeichnis (OUTDIR) und Dateinamen (GNAME) vergeben;
* Nach dem Kopieren des Programmcodes gegebenenfalls die Ausrichtung der Spalten im Datalines-Statement
justieren (Datensatz ksfedat);

%LET OUTDIR=C:\temp;
%LET GNAME=KSFE2019_map1.svg;

*****;
* Angaben zu KSFE-Veranstaltungsorten;
data ksfedat;
input @1 nr 2. @4 jahr 4. @8 stadt $13. @22 ort $43. @67 plz $5. @73 y 10.7 @84 x 10.7;
datalines;
22 2018 Mannheim      Universitätsklinikum Mannheim      68167 49.5007112 8.4922842
21 2017 Krefeld       Fachhochschule Krefeld              47805 51.3176588 6.5635646
20 2016 Greifswald    Universität Greifswald              17489 54.0838668 13.3599784
19 2015 Hannover     Medizinische Hochschule Hannover    30625 52.3706369 9.8048107
18 2014 Göttingen    Universität Göttingen               37073 51.5341276 9.9354167
17 2013 Ulm           Universität Ulm                      89081 48.3888754 9.9516668
16 2012 Dresden      Technische Universität Dresden       01069 51.0373465 13.734986
15 2011 Heidelberg   Ruprecht-Karls-Universität Heidelberg 69117 49.4109412 8.7069419
14 2010 Berlin       Freie Universität Berlin             14195 52.4516330 13.2712468
13 2009 Halle        Martin-Luther-Universität Halle-Wittenberg 06108 51.4904397 11.9628078
12 2008 Aachen       Universitätsklinikum der RWTH Aachen 52074 50.7662304 6.0419539
11 2007 Ulm          Universität Ulm                      89081 48.3888754 9.9516668
10 2006 Hamburg      Universität Hamburg                 20146 53.5503    10.0007000
9 2005 Berlin        Humboldt-Universität zu Berlin      12489 52.4359806 13.5348303
8 2004 Schmalkalden Fachhochschule Schmalkalden         98574 50.7171567 10.442868
7 2003 Potsdam      Universität Potsdam                 14469 52.4003345 13.0095665
6 2002 Dortmund     Universität Dortmund                44227 51.4646761 7.4152191
5 2001 Hohenheim    Universität Hohenheim                70599 48.712203  9.207501
4 2000 Gießen       Justus-Liebig-Universität Gießen    35390 50.5865732 8.6810508
3 1999 Heidelberg   Ruprecht-Karls-Universität Heidelberg 69117 49.4109412 8.7069419
2 1998 Jena         Friedrich-Schiller-Universität Jena  07743 50.948187  11.5946335
1 1997 Berlin       Humboldt-Universität zu Berlin      10117 52.5227915 13.3914374
;
run;

proc sort data=ksfedat;
by stadt jahr;
run;

* Anzahl der KSFEs pro Stadt zählen und Texte für Karte vorbereiten;
data ksfedat;
format jahre $25.;
retain num jahre;
set ksfedat;
by stadt;
if first.stadt then do;
    num=1;
    jahre=compress(jahr);
end;
else do;
    num+1;
    jahre=strip(jahre)||", "||compress(jahr);
end;
if last.stadt then do;
    jahre="("||strip(jahre)||")";
    output;
end;
run;

* Werte für die Einzeichnung in der Karte projektieren;
proc gproject data=ksfedat out=dot parmin=mapsgfk.projparm parmentry=germany;
id stadt;
```

```

run;

*****;
* Annotate Datensatz für Einzeichnung von Punkten, Städtename und Jahreszahlen erstellen;
data anno;
length function style color $20 text $25;
retain xsys ysys '2' hsys '1' when 'a';
set dot (in=a) dot (in=b) dot (in=c);
xsys=2;
ysys=2;
hsys=1;
when='a';
if a then do; * Punkte einzeichnen;
    function='pie';
    rotate=360;
    size=0.70*num;
    style='psolid';
    color="#006aa3";
end;
if b then do; * Städtenamen einzeichnen;
    function='label';
    color="#404040";
    if compress(stadt) not in('Potsdam','Mannheim') then do;
        position='6';
        if num=1 then text="  ||compress(stadt);
        if num=2 then text="   ||compress(stadt);
        if num=3 then text="    ||compress(stadt);
    end;
    else do;
        position='4';
        text=compress(stadt);
        x=x-0.001;
    end;
    style="Arial";
    size=2;
end;
if c then do; * Jahreszahlen;
    function='label';
    color="#404040";
    if compress(stadt) not in('Potsdam','Mannheim') then do;
        position='9';
        if num=1 then text="  ||strip(jahre);
        if num=2 then text="   ||strip(Jahre);
        if num=3 then text="    ||strip(Jahre);
    end;
    else do;
        text=strip(jahre);
        position='7';
        x=x-0.001;
    end;
    style="Arial";
    size=1.5;
end;
if (b or c) and compress(stadt) in('Jena','Dortmund') then y=y+0.001; * Einträge justieren;
run;

*****;
* Zusatzinformation für Mouse-Over-Funktion;

data info;
set mapsgfk.germany_attr (keep=id idname);
format infotxt $200.;
infotxt='ALT="||substr(id,4,5)||' '||strip(idname)||"';
run;

*****;
* Maps;

* Karte mit Bundeslandgrenzen;
proc gremove data=mapsgfk.germany out=map;
by id1;
id id;
run;

data map1;
set map;
dummy=1;
run;

```

```
* Karte Kreisgrenzen;
data mapk1;
merge mapsgfk.germany info;
by id;
dummy=1;
run;

*****;
* Ausgabe;

ods _all_ close;

options printerpath=SVGT nodate nonumber nofontembedding;

ods printer file="%OUTDIR.\&GNAME" dpi=200;

goptions reset=all ftitle="arial" cback="CXFFFFFF";

title "KSFE Veranstaltungsorte 1997 bis 2018";

pattern value=ms color="CXEBEBEB" repeat=50;

proc gmap data=mapk1 map=mapk1;
id id;
choro dummy / statistic=first coutline="CXDEDEDE" woutline=1 nolegend html=infotxt;
run;
quit;

pattern value=e; * keine Musterfüllung;

proc gmap data=map1 map=map;
id id1;
choro dummy / statistic=first coutline="CXBFBFBF" woutline=1 nolegend anno=anno;
run;
quit;

title;

ods printer close;
ods html;
```